PŘÍKLAD 1: Najděte polynomickou funkci, jejíž graf prochází body [1, 2], [−4, 7] a [6, −13].

```
> restart;
> with(linalg):
> A:=matrix([[1,1,1],[16,-4,1],[36,6,1]]);
  B:=matrix([[2],[7],[-13]]);
```

$$A := \begin{bmatrix} 1 & 1 & 1 \\ 16 & -4 & 1 \\ 36 & 6 & 1 \end{bmatrix}$$

$$B := \begin{bmatrix} 2 \\ 7 \\ -13 \end{bmatrix}$$

```
> Ai:=inverse(A);
```

$$\mathrm{Ai} := \begin{bmatrix} \frac{-1}{25} & \frac{1}{50} & \frac{1}{50} \\ \frac{2}{25} & \frac{-7}{50} & \frac{3}{50} \\ \frac{24}{25} & \frac{3}{25} & \frac{-2}{25} \end{bmatrix}$$

```
> K:=evalm(Ai.B);
```

$$K := \begin{bmatrix} \frac{-1}{5} \\ \frac{-8}{5} \\ \frac{19}{5} \end{bmatrix}$$

```
> f:=unapply(K[1,1]*x^2+K[2,1]*x+K[3,1],x);
```

$$f := x \rightarrow -\frac{1}{5} x^2 - \frac{8}{5} x + \frac{19}{5}$$

```
> Body:=[[1,2],[-4,7],[6,-13]];
```

$$Body := [[1, 2], [-4, 7], [6, -13]]$$

```
> BodyG:=plot(Body,style=point,symbol=cross,symbolsize=30,color=re
  d):
> PolynomG:=plot(f(x),x,color=blue):
> plots[display](PolynomG,BodyG);
```



Míra podmíněnosti matice (míra její vhodnosti pro numerické řešení) $\kappa(A)$:

```
> norm(A,1); norm(Ai,1);
```

$$53$$

$$\frac{27}{25}$$

```
> kappa(A):=evalf(norm(A,1)*norm(Ai,1));
```

$$\kappa(A) := 57.24000000$$

Matice A je Vandermondova matice pro hodnoty x_i; $x_1 = 1$, $x_2 = -4$, $x_3 = 6$.

```
> LinearAlgebra[VandermondeMatrix]([1,-4,6]);
```

$$\begin{bmatrix} 1 & 1 & 1 \\ 1 & -4 & 16 \\ 1 & 6 & 36 \end{bmatrix}$$

PŘÍKLAD 2: Proveďte interpolaci pěti bodů [−63, 52], [−57, 83], [−78, −79], [−12, −25], [−45, 83] polynomem čtvrtého stupně.

```
> A:=LinearAlgebra[VandermondeMatrix]([-63,-57,-78,-12,-45]);
```

$$A := \begin{bmatrix} 1 & -63 & 3969 & -250047 & 15752961 \\ 1 & -57 & 3249 & -185193 & 10556001 \\ 1 & -78 & 6084 & -474552 & 37015056 \\ 1 & -12 & 144 & -1728 & 20736 \\ 1 & -45 & 2025 & -91125 & 4100625 \end{bmatrix}$$

Míra podmíněnosti matice je vysoká. Při numerickém řešení může dojít k závažným chybám ve výsledných hodnotách.

```
> kappa(A):=evalf(norm(A,1)*norm(inverse(A),1));
```

$$\kappa\left(\begin{bmatrix} 1 & -63 & 3969 & -250047 & 15752961 \\ 1 & -57 & 3249 & -185193 & 10556001 \\ 1 & -78 & 6084 & -474552 & 37015056 \\ 1 & -12 & 144 & -1728 & 20736 \\ 1 & -45 & 2025 & -91125 & 4100625 \end{bmatrix}\right) := 0.2997064875\ 10^{10}$$

PŘÍKLAD 3: Ukázka špatně podmíněné soustavy (matice) [převzato z http://www.kvd.zcu.cz/materialy/numet/_numet.htm]
Vidíme, že malé změny v koeficientech rovnic způsobily velké změny v řešení.

```
> r1_1:=2*x+6*y=8; r2_1:=2*x+6.00001*y=8.00001;
```

$$r1\_1 := 2\,x + 6\,y = 8$$

$$r2\_1 := 2\,x + 6.00001\,y = 8.00001$$

```
> Res_Soust_1:=solve([r1_1,r2_1],[x,y]);
```

$$Res\_Soust\_1 := [[x = 1., y = 1.]]$$

```
> r1_2:=2*x+6*y=8; r2_2:=2*x+5.99999*y=8.00002;
```

$$r1\_2 := 2\,x + 6\,y = 8$$

$$r2\_2 := 2\,x + 5.99999\,y = 8.00002$$

```
> Res_Soust_2:=solve([r1_2,r2_2],[x,y]);
```

$$Res\_Soust\_2 := [[x = 10., y = -2.]]$$

```
> A1:=linalg[genmatrix]([r1_1,r2_1],[x,y]);
```

$$A1 := \begin{bmatrix} 2 & 6 \\ 2 & 6.00001 \end{bmatrix}$$

```
> kappa(A1):=evalf(norm(A1,1)*norm(inverse(A1),1));
```

$$\kappa(A1) := 0.4800010000\;10^{7}$$

```
> A2:=linalg[genmatrix]([r1_2,r2_2],[x,y]);
```

$$A2 := \begin{bmatrix} 2 & 6 \\ 2 & 5.99999 \end{bmatrix}$$

```
> kappa(A2):=evalf(norm(A2,1)*norm(inverse(A2),1));
```

$$\kappa(A2) := 0.4799996000\;10^{7}$$

Užití Lagrangeova interpolačního polynomu k vyřešení PŘÍKLADU 1:

```
> L0:=(x-x1)*(x-x2)/((x0-x1)*(x0-x2));
  L1:=(x-x2)*(x-x0)/((x1-x2)*(x1-x0));
  L2:=(x-x0)*(x-x1)/((x1-x2)*(x1-x0));
```

$$L0 := \frac{(x - x1)\,(x - x2)}{(x0 - x1)\,(x0 - x2)}$$

$$L1 := \frac{(x - x2)\,(x - x0)}{(x1 - x2)\,(x1 - x0)}$$

$$L2 := \frac{(x - x0)\,(x - x1)}{(x1 - x2)\,(x1 - x0)}$$

```
> x0:=1: x1:=-4: x2:=6: y0:=2: y1:=7: y2:=-13:
> L0; L1; L2;
```

$$-\frac{(x + 4)\,(x - 6)}{25}$$

$$\frac{(x-6)(x-1)}{50}$$

$$\frac{(x-1)(x+4)}{50}$$

```
> f:=unapply(simplify(L0*y0+L1*y1+L2*y2),x);
```

$$f := x \rightarrow -\frac{1}{5}x^2 - \frac{8}{5}x + \frac{19}{5}$$

PŘÍKLAD 4: Ukázka toho, jak Lagrangeův interpolační polynom osciluje mezi danými body.
Interpolujme užitím Lagrangeova polynomu (2., 4. a 10. stupne) funkci

$\mathrm{f}(x) = \frac{1}{1+x^2}$:

```
> with(CurveFitting):
> f:=x->1/(1+x^2);
```

$$f := x \rightarrow \frac{1}{1+x^2}$$

```
> L2:=PolynomialInterpolation([[-3,f(-3)],[0,f(0)],[3,f(3)]], x);
```

$$L2 := -\frac{x^2}{10} + 1$$

```
> L2:=PolynomialInterpolation([[-3,f(-3)],[0,f(0)],[3,f(3)]],
  x,form=Lagrange);
```

$$L2 := \frac{x(x-3)}{180} - \frac{(x+3)(x-3)}{9} + \frac{(x+3)x}{180}$$

```
> L2:=PolynomialInterpolation([[-3,f(-3)],[0,f(0)],[3,f(3)]],
  x,form=Newton);
```

$$L2 := \left(-\frac{x}{10} + \frac{3}{10}\right)(x+3) + \frac{1}{10}$$

```
> L4:=PolynomialInterpolation([[-3,f(-3)],[-1.5,f(-1.5)],[0,f(0)],
  [1.5,f(1.5)],[3,f(3)]], x);
```

$$L4 := 0.03076923076\, x^4 + 0.2\, 10^{-10}\, x^3 - 0.3769230767\, x^2 + 0.4\, 10^{-9}\, x + 1.000000000$$

```
> L4:=PolynomialInterpolation([[-3,f(-3)],[-1.5,f(-1.5)],[0,f(0)],
  [1.5,f(1.5)],[3,f(3)]], x,form=Lagrange);
```

$$\begin{aligned} L4 := {} & 0.0008230452675\,(x+1.5)\,x\,(x-1.5)\,(x-3) \\ & - 0.01012978791\,(x+3)\,x\,(x-1.5)\,(x-3) \\ & + 0.04938271605\,(x+3)\,(x+1.5)\,(x-1.5)\,(x-3) \\ & - 0.01012978791\,(x+3)\,(x+1.5)\,x\,(x-3) \\ & + 0.0008230452675\,(x+3)\,(x+1.5)\,x\,(x-1.5) \end{aligned}$$

```
> L4:=PolynomialInterpolation([[-3,f(-3)],[-1.5,f(-1.5)],[0,f(0)],
  [1.5,f(1.5)],[3,f(3)]], x,form=Newton);
```

$$L4 := (((0.03076923076\,x - 0.1384615384)\,x + 0.1076923077)\,(x + 1.5) + 0.1384615385)\,(x + 3) + \frac{1}{10}$$

```
> L10:=PolynomialInterpolation([[-3,f(-3)],[-2.4,f(-2.4)],[-1.8,f(
  -1.8)],[-1.2,f(-1.2)],[-0.6,f(-0.6)],[0,f(0)],[0.6,f(0.6)],[1.2,
  f(1.2)],[1.8,f(1.8)],[2.4,f(2.4)],[3,f(3)]], x);
```

$$L10 := -0.001051377579\,x^{10} + 0.27\;10^{-10}\,x^{9} + 0.02186865370\,x^{8} - 0.43\;10^{-9}\,x^{7} - 0.1612611352\,x^{6} + 0.15\;10^{-8}\,x^{5} + 0.5362718757\,x^{4} - 0.24\;10^{-8}\,x^{3} - 0.9084551971\,x^{2} + 0.2\;10^{-9}\,x + 1.000000001$$

```
> plot([f(x),L2,L4,L10],x=-3..3,y=-0.2..1,color=[red,blue,green,ma
  genta],thickness=5);
```




```
>
>
>
>
```